********Table O.1 - Indonesia********

eststo clear

*Load data and create differences
cd "$reploc/data/created_data"
use indo_full, clear

sort pidlink ifls
drop if ifls==.
keep if panel==1


*Generate age dummies, and rename to avoid confusion
tab imp_age, gen(impagedum)

foreach var of varlist impagedum* {
local nam1=substr("`var'",10,.)
rename `var' impage`nam1'dum
}

keep rbuc* gambleaverse ifls pidlink inflation imp_age panel biryear_impl grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co? grbrprovSD_lam0co? birprov_impl impage*dum

local varsdumsage
foreach var of varlist impage*dum {
local varsdumsage `varsdumsage' `var'
}

reshape wide rbuc*  gambleaverse inflation imp_age panel grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co? grbrprovSD_lam0co? `varsdumsage', i(pidlink) j(ifls)


local varlist1 rbuc rbucB rbuc_nga rbuc_garefined rbucgaref1a rbucgaref2a rbucgaref3a rbucgaref4a gambleaverse inflation grbrprovAvg_lam0 grbrprovSD_lam0 grbrprovAvg_lam0co2 grbrprovSD_lam0co2 grbrprovAvg_lam0co3 grbrprovSD_lam0co3 `varsdumsage'


foreach var in `varlist1'  {
gen `var'dif=.
replace `var'dif=`var'5-`var'4
}

*Generate cohort by birthyear variables to cluster on
egen coh_prov_clust=group(biryear_impl birprov_impl)

*Regressions
local x rbuc_garefined
local rhsvari oldindoonly
local rhsvaruse _lam0co3dif
local usei grbr

drop if `x'dif==.



*Column 1
eststo:  reg `x'dif  `usei'provAvg`rhsvaruse'  inflationdif impage*dumdif, vce(cluster coh_prov_clust)

*Column 2
eststo:  reg `x'dif `usei'provSD`rhsvaruse'  inflationdif impage*dumdif, vce(cluster coh_prov_clust)

*Column 3
eststo:  reg `x'dif  `usei'provAvg`rhsvaruse' `usei'provSD`rhsvaruse'  inflationdif impage*dumdif, vce(cluster coh_prov_clust)


cd "$reploc/results"	
esttab using TableO1_Indonesia.tex, star(* 0.1 ** 0.05 *** 0.01) b(2) se(2) replace
